EMailAdressen aus einem Ordner per VBA auslesen
Versionen: Outlook 2007, 2003, 2002/XP und 2000

Nehmen wir an, dass sich in einem Ordner E-Mails befinden, aus denen Sie die Adressen der Absender auslesen und zur weiteren Verarbeitung auslesen und speichern möchten. Um das manuelle Abtippen zu vermeiden, müssen sie auf jeden Fall ein VBA-Makro anlegen. Allerdings ist es nur in Outlook 2007 und 2003 möglich, die E-Mail-Adresse eines Absenders ohne Umwege auszulesen.

In Outlook 2000 und Outlook 2002/XP geht das nur über die so genannten CDO (Collaboration Data Objects), die normalerweise automatisch mit Outlook installiert werden, sodass Sie aus einem VBA-Makro auf die entsprechenden Eigenschaften und Methoden zugreifen können.

Der folgende Programmcode demonstriert die generelle Vorgehensweise. Damit überprüfen Sie alle Elemente in einem von Ihnen ausgewählten Ordner (Outlook 2007 und 2003) bzw. im Posteingang (Outlook 2002/XP und 2000) und schreiben jeweils den Absendernamen sowie seine E-Mail-Adresse in das Direktfenster des Visual Basic-Editors. Das ist zwar nicht sonderlich komfortabel, aber es zeigt, wie Sie an die gewünschten Informationen kommen. Sie können den Code nach Belieben erweitern, um Namen und Adressen zum Beispiel als Kontaktdaten in einem anderen Outlook-Ordner zu speichern.

Den Beispielcode geben Sie folgendermaßen ein:

· Starten Sie Outlook und wechseln Sie mit der Tastenkombination Alt+F11 in den Visual Basic-Editor.

· Sorgen Sie dafür, dass das Direktfenster sichtbar ist, indem Sie Ansicht-Direktfenster anwählen.

· Rufen Sie Einfügen-Modul auf.

· Geben Sie in das daraufhin geöffnete Codefenster den Prozedurcode ein:

Sub ReadAdrFromFolder()
  'Outlook
  Dim olNameSpace As NameSpace
  Dim olFolder As MAPIFolder
  Dim olItem As Object

  'CDO
  Dim cdoSession As Object
  Dim cdoFolder As Object
  Dim cdoMsg As Object
  Dim cdoSender As Object

  On Error Resume Next
  If Val(Application.Version) >= 11 Then '2003, 2007
    Set olNameSpace = Application.GetNamespace("MAPI")
    Set olFolder = olNameSpace.PickFolder

'Abbrechen geklickt
    If olFolder Is Nothing Then Exit Sub 

    For Each olItem In olFolder.Items
      Debug.Print olItem.SenderName & _
          " <" & olItem.SenderEmailAddress & ">"
    Next olItem
    On Error GoTo 0
    Set olItem = Nothing
    Set olFolder = Nothing
    Set olNameSpace = Nothing
  Else 'Outlook 2000, 2002/XP
    Set cdoSession = CreateObject("MAPI.Session")
    cdoSession.Logon "", "", False, False
    Set cdoFolder = cdoSession.Inbox
    For Each cdoMsg In cdoFolder.Messages
      Set cdoSender = cdoMsg.Sender
      Debug.Print cdoSender.Name & _
          " <" & cdoSender.Address & ">"
    Next cdoMsg
    On Error GoTo 0
    Set cdoMsg = Nothing
    Set cdoSender = Nothing
    Set cdoFolder = Nothing
    cdoSession.Logoff
    Set cdoSession = Nothing
  End If
End Sub
· Testen Sie den Programmcode durch Aufruf von Extras-Makros und anschließendem Doppelklick auf den Makronamen "ReadAdrFromFolder".

· Sämtliche Absendernamen und -adressen aus dem Posteingang erscheinen anschließend im Direktfenster.

· Wenn Sie den VB-Editor wieder schließen wollen, öffnen Sie das Menü Datei und rufen Sie Schließen und zurück zu Microsoft (Office) Outlook auf.

Kurz zur Erläuterung des Programmcodes: Damit die Prozedur in allen Outlook-Versionen lauffähig ist, fragt sie zuerst die Versionsnummer ab. Nur in Outlook 2007 und 2003 mit einer Versionsnummer größer "11" können Sie anschließend auf die Standard-Objekte von Outlook zurückgreifen. So erstellen Sie Objektvariablen für den Ordner ("Set olFolder = olNameSpace.PickFolder") und die darin enthaltenen Elemente, die Sie dann einzeln abarbeiten. Für jedes Element lassen Sie "SenderName" und "SenderEmailAddress" per "Debug.Print" im Direktbereich ausgeben.

Die Eigenschaft "SenderEmailAddress" gab es in früheren Outlook-Versionen noch nicht. Deshalb können Sie die Absenderadresse nur per CDO ermitteln. Dafür initiieren Sie zuerst eine MAPI-Sitzung:

Set cdoSession = CreateObject("MAPI.Session")
cdoSession.Logon "", "", False, False

Danach durchlaufen Sie auch hier alle Elemente im Posteingang ("cdoSession.Inbox"), wobei Sie jeweils auf das "Sender"-Objekt zugreifen. Anzeigename und E-Mail-Adresse erhalten Sie mit den Eigenschaften "Name" und "Address".

